
24. Interference

General Description

This program will prove invaluable to physics teachers trying to
explain the thorny problems of interference to A-level classes.
The program produces two sets of overlapping semi-circular
waves as a simulation of two-slit diffraction. The simulation
produces clearly identifiable interference patterns which can be
used in parallel with verbal explanation. The user instructions go
to some lengths to point students in the right direction, and ask
the user to supply sample values for slit operation and wave-
length.

Detailed Description

Lines 60-380 Main Program.
80-160 Instructions.
170-230 Draw slits at required separation and display data

supplied. Incident beam and diffracted waves are drawn. If
requested, interference pattern is emphasised.

240-380 Run program again? Line 320, INKEY waits 10 secs
before returning null string.

420-680 PROChelp: This explains the program's facilities.
700-820 PROC data: This requests user data. D% = slit

separation. L% = wavelength. The scaling factor of 2 is
introduced so that the values of wavelength may be more
realistic for visible light.

840-990 PROCslits: Slits are drawn using command
PLOT85.

1010-1130 PROCbeam: The incident beam is assumed to be a
parallel one. The spacing between waves is L% and the position
of the waves is calculated to fit with the first diffracted wavelet.

1030 calculates position of first wave, 1040 the number of waves
that will fit in available space.

1150-1370 PROCwaves: Waves from left-hand slit drawn by
1200-1280. Right-hand slit waves are drawn 1290-1360.

1390-1530 PROCrerun: Another set of waves to be super-
imposed on the first?

1540-1690 PROCmaxima: This procedure draws lines in the
direction of constructive interference.

Educational Notes

In the classroom, this simulation could effectively replace
existing audio-visual demonstrations. If used with A-Level
groups (with a well-adjusted TV monitor), measurements of
angle can be made directly from the screen to verify (or
otherwise) the students' predictions.

For the best results, I would suggest a slit separation of less
than 400mm and wavelengths between 100 and 350mm. For
example D%=300, L%=230mm.

Program Listing

   10   REM ********************
   20   REM ** "INTERFERENCE" **
   30   REM ** by ALAN BUGH   **
   40   REM **   FEB. 1983    **
   50   REM ********************
   60  MODE7
   70 VDU23;8202;0;0;0;:REM CURSOR OFF.
   80 PRINTTAB(11,12)CHR$(141);CHR$(133);"INTERFERENCE"
   90 PRINTTAB(11,13)CHR$(141);CHR$(133);"INTERFERENCE"
  100 TIME=0:REPEAT:UNTIL TIME>500
  110 MODE1
  120 PROChelp
  130 MODE1
  140 VDU23;8202;0;0;0;
  150 PROCdata
  160 Q$="N"
  170 PROCslits
  180 PRINTTAB(0,25)"d =";D%*2;"nm"
  190 PRINTTAB(0,27)"L1=";L%*2;"nm"
  200 GCOL0,1
  210 PROCbeam
  220 PROCwaves
  230 PROCmaxima
  240 PROCrerun
  250 IF Q$="Y" THEN GOTO210: REM RERUN
  260 PRINTTAB(9,29)"Do you want to start again?":*FX15,1
  270 E$=GET$
  280 IF E$="Y" THEN GOTO 130

  290 CLS
  300 PRINTTAB(5,16)"You have ten seconds to change"
  310 PRINTTAB(5,17)"your mind!-----Any key will do.":*FX15
  320 Z$=INKEY$(1000)
  330 IF Z$=""GOTO340ELSE 110
  340 MODE7
  350 PRINTTAB(14,12)CHR$(141);CHR$(133);"GOODBYE!"
  360 PRINTTAB(14,13)CHR$(141);CHR$(133);"GOODBYE!"
  370 VDU30
  380 END
  390 REM*******************************
  400 REM END OF MAIN PROGRAM
  410 REM ******************************
  420 DEFPROChelp
  430 REM********************************
  440 REM PROVIDES INFORMATION ABOUT THE PROGRAM IF REQUESTE
D.
  450 REM********************************
  460 PRINTTAB(7,16)"Do you want instructions?":*FX15,1
  470 I$=GET$
  480 IF I$="N" THEN ENDPROC
  490 CLS
  500 PRINTTAB(6,5)"This program produce a graphical"
  510 PRINT"representation  of  an  interference"
  520 PRINT"pattern produced by a pair of coherent"
  530 PRINT"light sources."
  540 PRINT:PRINTSPC(6)"You will have to  supply values"
  550 PRINT"of slit separation between 60nm and"
  560 PRINT"1280nm, and of wavelength between"
  570 PRINT"60nm and 600nm."
  580 PRINT:PRINTSPC(6)"YOU MAY VIEW  THE  WAVES PRODUCED"
  590 PRINT"FROM EACH SLIT ALTERNATELY OR ONE SLIT"
  600 PRINT"AT A TIME."
  610 PRINTTAB(6,19)"If you wish to study the effect"
  620 PRINT"on the pattern of changing the"
  630 PRINT"wavelength of the light, you  will  be"

  640 PRINT"given the opportunity at the end of"
  650 PRINT"the first run."
  660 PRINTTAB(8,30) "PRESS ANY KEY TO CONTINUE":*FX15,1
  670 Z$=GET$
  680 ENDPROC
  690 REM *****************************
  700 DEFPROCdata
  710 REM REQUESTS SAMPLE VALUES FROM USER.
  720 REM******************************
  730 INPUTTAB(0,5)"Slit operation, in nm, "D%:*FX15,1
  740 D%=D%/2
  750 IFD%>640 OR D%<30 THENPRINT"OUT OF RANGE. (60 TO 1280)
":VDU7:TIME=0:REPEAT:UNTILTIME>200:PRINTTAB(0,5)SPC(80):GOTO
730
  760 INPUTTAB(0,7)"Wavelength of light, nm, ="L%:*FX15,1
  770 L%=L%/2
  780 IFL%>300 OR L% <30 THEN PRINT"OUT OF RANGE. (60 TO 600
)":VDU7:TIME=0:REPEAT:UNTILTIME>200:PRINTTAB(0,7)SPC(80):GOT
O760
  790 PRINTTAB(0,9)"Do you want the waves to be drawn from e
ach slit alternately? (Y/N)":*FX15,1
  800 FLAG$=GET$
  810 CLS
  820 ENDPROC
  830 REM ******************************
  840 DEFPROCslits
  850 REM DRAWS THE SLITS AT SELECTED SEPARATION.
  860 REM*******************************
  870 MOVE 0,280
  880 DRAW 640-(D%/2+10),280
  890 PLOT 85,0,300
  900 PLOT85,640-(D%/2+10),300

  910 MOVE 640-(D%/2-10),280
  920 DRAW 640+(D%/2-10),280
  930 PLOT85,640-(D%/2-10),300
  940 PLOT85,640+(D%/2-10),300
  950 MOVE 640+D%/2+10,280
  960 DRAW 1279,280
  970 PLOT85,640+D%/2+10,300
  980 PLOT85,1279,300
  990 ENDPROC
 1000 REM *******************************
 1010 DEFPROCbeam
 1020 REM DRAWS INCIDENT BEAM AT BOTTOM OF THE SCREEN
 1025 REM****************************
 1030 START%=(300 MOD L% +10)
 1040 NUMBER%=300 DIV L%
 1050 FOR P =1 TO NUMBER%
 1060   MOVE300,START%
 1070   DRAW300+800,START%
 1080   START%=START%+L%
 1090   TIME=0
 1100   REPEAT
 1110   UNTIL TIME>50
 1120 NEXTP
 1130 ENDPROC
 1140 REM******************************
 1150 DEFPROCwaves
 1160 REM DRAWS THE ADVANCING WAVEFRONT FROM EACH SLIT.
 1170 REM******************************
 1180 C1%=640-(D%/2)
 1190 C2%=640+(D%/2)
 1200 FOR R%=10 TO 700 STEP L%
 1210   MOVEC1%+R%,300
 1220   FOR A=0 TO PI STEP PI/20
 1230     X1=C1%+(R%*COS(A))
 1240     Y=300+(R%*SIN(A))
 1250     PLOT5,X1,Y
 1260   NEXTA
 1270   IF FLAG$="Y" THEN GOTO1300: REM ALTERNATE?
 1280 NEXTR%
 1290 FOR R%=10 TO 700 STEP L%
 1300   MOVEC2%+R%,300
 1310   FOR A=0 TO PI STEP PI/20

 1320     X2=C2%+(R%*COS(A))
 1330     Y=300+(R%*SIN(A))
 1340     PLOT5,X2,Y
 1350   NEXTA
 1360 NEXTR%
 1370 ENDPROC
 1380 REM ******************************
 1390 DEFPROCrerun
 1400 REM ENQUIRES IF THE USER WISHES TO CONTINUE
 1410 REM******************************
 1420 PRINTTAB(0,29)SPC(79):IF Q$="N"THEN PRINTTAB(0,29)"Do 
you want to SUPERIMPOSE a second wave-length? (Y/N)":*FX15,1
 1430 IFQ$="Y"THEN Q$="N":ENDPROC
 1440 Q$=GET$
 1450 PRINTTAB(0,29)SPC(79)
 1460 IF Q$="Y"THEN INPUTTAB(0,29)"Second wavelength, in nm,
 ="L2%ELSE ENDPROC
 1470 L%=L2%/2
 1480 IFL%>300 OR L% <30 THEN PRINT"OUT OF RANGE. (60 TO 600
)":VDU7:TIME=0:REPEAT:UNTILTIME>200:GOTO1450
 1490 PRINTTAB(0,29)SPC(79)
 1500 VDU19,2,6,0,0,0
 1520 GCOL0,2
 1530 ENDPROC
 1535 REM*******************************
 1540 DEFPROCmaxima:*FX15,1

 1545 REM USES nL=dsinA TO CALCULATE THE DIRECTIONS OF CONST
RUCTIVE INTERFERENCE.
 1546 REM******************************
 1550 PRINTTAB(0,29)SPC(79)
 1560 PRINTTAB(10,29)"EMPHASISE INTERFERENCE?"
 1570 I$=GET$:IF I$="N" THEN ENDPROC
 1580 N%=0
 1590 REPEAT
 1600   B=N%*(L%/D%)
 1610   IFB>1 GOTO 1680
 1620   N%=N%+1
 1630   A=ASN(B)
 1640   MOVE640,300
 1650   MOVE640+(700*B),300+(700*COS(A))
 1660   MOVE640,300
 1670   DRAW640-(700*B),300+(700*COS(A))
 1680 UNTILB>1
 1690 ENDPROC

